
METR 4433 – Mesoscale Meteorology 
 

Problem Set #3 
 

Sample Answers 
 
1. (50%) A vertical wind profile is given by the following table: 
 
 z (height, km)  direction, deg) V(speed, m/s) 
 0 120 5 
 1 150 10 
 2 180 15 
 3 220 20 
 4 250 25 
 5 270 30 
 6 310 40 
 
Assume that the storm motion vector is from 225 degrees (from SW) and the speed is 12 
m/s. 
 
a). Plot the hodograph and the storm-relative velocity vectors at each level 
 
b). Calculate the horizontal vorticity (vector, in terms of the vorticity components or in 

magnitude and direction) in each of the six layers between the levels of observations 
 
c). Determine the mean (storm-relative) wind vector in each of these size layers 
 
d). Using the layer-mean wind obtained above, calculate the storm-relative helicity in 

each of the six layers, and determine the vertically integrated environmental helicity 
in the lowest three kilometers 

 
e) Calculate the (storm-relative) streamwise vorticity and (storm-relative) relative helicity 

in each of the six layers 
 
f). Discuss your results and their significance in terms of the their effect on the behavior 

and type of the storms that occur in such environment 
 
g). For this wind profile, what kind of CAPE values will give you a BRN that suggests a 

high probability of multicell and supercell storms, respectively? 
 
 
The following is a Fortran 90 program to calculate the above quantities, and to produced 

the hodograph. 
 
PROGRAM HELICITY 
! 
! Program to calculate parameters of Mesoscale Meteorology Homework #2 

! It calls ZXPLOT (http://www.caps.ou.edu/ZXPLOT) graphics library to  
! for plotting. 
! Written by Ming Xue 
! 5/5/2002 
! 
  IMPLICIT NONE 
  INTEGER, PARAMETER :: n = 6 
  REAL :: speed(0:n),direction(0:n),z(0:n) 
  REAL :: ua(0:n), va(0:n), ur(0:n), vr(0:n)  
                                 ! u-v components of abs and rel. velocity 
 
  REAL :: ustorm, vstorm, Cstorm,dirstorm ! storm motion vector 
  REAL :: omegax(n),omegay(n) ! horizontal vorticity 
  REAL :: H(n)     ! Storm-relative helicity 
  REAL :: um(n), vm(n)   ! layer mean velocity 
  REAL :: h3km 
  REAL :: omegas(n) ! streamwise vorticity components 
  REAL :: RH(n)     ! relative helicity 
  REAL CAPE, BRN, S, u6km,v6km,u500m,v500m 
  INTEGER i 
  REAL :: alpha, pi, deg2rad 
 
  data z/ 0.0, 1000.0, 2000.0, 3000.0, 4000.0, 5000.0, 6000.0/ 
  data speed/5.0,10.0,15.0,20.0,25.0,30.0,40.0/  
  data direction/120.0,150.0,180.0,220.0,250.0,270.0,310.0/ 
 
  CALL XDEVIC 
  CALL XPSPAC(0.1,0.9, 0.1, 0.9) 
  CALL XMAP (-40.0, 40.0, -40.0, 40.0) 
  CALL XAXFMT('(I3)') 
  CALL XAXES(0.0, 5.0, 0.0, 5.0) 
 
 
  CALL Xdash 
  DO i=5,40,5 
    CALL Circle(0.0,0.0,float(i)) 
  ENDDO 
  CALL xfull 
 
  deg2rad = atan(1.0)/45.0 
 
  cstorm = 12.0 
  dirstorm = 225.0 
 
  alpha = 360.0-(dirstorm-180.0-90.0) 
  ustorm = cstorm*cos(alpha*deg2rad) 
  vstorm = cstorm*sin(alpha*deg2rad) 
 
  Print*,'u_storm = ', ustorm 
  Print*,'v_strom = ', vstorm 
 
  CALL XPENUP(0.0, 0.0) 
  CALL XPENDN(ustorm, vstorm) 
! 
! Determine Cartesian components of the absolute velocity  
! at each observation level  
! 
  DO i=0,n 
    alpha = 360.0-(direction(i)-180.0-90.0)  
   ! Convert to polar angle from x-axis in counterclock wise direction 
 
    ua(i) = speed(i) * cos(alpha*deg2rad) 
    va(i) = speed(i) * sin(alpha*deg2rad) 

    CALL XPENUP(0.0, 0.0) 
    CALL XPENDN(ua(i),va(i)) 
  ENDDO 
 
  CALL XPENUP(ua(0),va(0)) 
  DO i=1,n 
    CALL XPENDN(ua(i),va(i)) 
  ENDDO 
! 
! Detemine storm-relative velocity at observation level 
! 
  DO i=0,n 
    ur(i) =ua(i)-ustorm 
    vr(i) =va(i)-vstorm 
    CALL XPENUP(ustorm,vstorm) 
    CALL XPENDN(ua(i),va(i)) 
  ENDDO 
!  
! Calculate horizontal vorticity in each layer 
! 
  DO i=1,n 
    omegax(i) = -(va(i)-va(i-1))/(z(i)-z(i-1)) 
    omegay(i) = +(ua(i)-ua(i-1))/(z(i)-z(i-1)) 
  ENDDO 
!  
! Calculate mean storm-relative velocity in each layer 
! 
  DO i=1,n 
    um(i) = 0.5*(ur(i)+ur(i-1))  
    vm(i) = 0.5*(vr(i)+vr(i-1))  
    CALL XPENUP(ustorm,vstorm) 
    CALL XPENDN(ustorm+um(i),vstorm+vm(i)) 
  ENDDO 
!  
! Calculate storm-relative helicity in each layer 
! H = V dot H, where V is the layer mean velocity 
! 
  DO i=1,n 
    H(i) = um(i)*omegax(i)+vm(i)*omegay(i)  
  ENDDO 
!  
! Vertically integrated storm-relative helicity in lowest 3km  
! 
  h3km = 0.0 
  DO i=1,3 
    h3km = h3km + h(i)*(z(i)-z(i-1)) 
  ENDDO 
!  
! Calculate streamwise vorticity and relative helicity in each layer  
! 
  DO i=1,n 
    omegas(i) = h(i)/sqrt(um(i)**2+vm(i)**2) 
    RH(i) = omegas(i)/sqrt(omegax(i)**2+omegay(i)**2) 
  ENDDO 
 
!  
! Determine mean wind in the 6 km layer for calculating BRN 
! We will use storm relative velocity here. Using abs. velocity 
! should give the same V6km - V500m. 
! 
  u6km = 0.0 
  v6km = 0.0 
  DO i=1,n 

    u6km=u6km+um(i) 
    v6km=v6km+vm(i) 
  ENDDO 
  u6km = u6km/n 
  v6km = v6km/n 
!  
! Determine mean wind in the lowest 500 m for calculating BRN 
! 
  u500m = 0.5*(ur(0)+ur(1))  ! find linearly interpolated value at 500m 
  v500m = 0.5*(vr(0)+vr(1))  ! find linearly interpolated value at 500m 
 
  u500m = 0.5*(ur(0)+u500m)  ! now mean velocity in the first 500m  
  v500m = 0.5*(vr(0)+v500m)  ! now mean velocity in the first 500m 
 
  S = sqrt( (u6km-u500m)**2+(v6km-v500m)**2 ) ! BRN shear 
 
  WRITE(6,'(/10a)')  & 
  '     Z(km)   Dir(deg)   V(m/s)    ua(m/s)   va(m/s)    ur(m/s)  vr(i)' 
  DO i=0,n 
  WRITE(6,'(7f10.3)')z(i)*0.001,direction(i),speed(i),ua(i),va(i),ur(i),vr(i) 
  ENDDO 
 
  WRITE(6,'(/10a)') & 
  '   Layer omegax(1/s) omegay(1/s)  um(m/s)  vm(m/s)    H(m/s**2)  RH     
OmegaS' 
  DO i=1,n 
  WRITE(6,'(i7,2F11.6,4f10.3,f11.6)') &  
  i,omegax(i),omegay(i),um(i),vm(i),h(i),rh(i),omegas(i) 
  ENDDO 
 
  WRITE(6,'(/3x,a,f10.3)') '3km integrated helicity (m**2/s**2)=', h3km 
  WRITE(6,'(3x,a,f10.3)') 'BRN shear S (m/s) =', S  
 
  WRITE(6,'(3x,a,f10.3)') 'CAPE for BRN=10 is ', 10*0.5*S**2 
  WRITE(6,'(3x,a,f10.3)') 'CAPE for BRN=45 is ', 45*0.5*S**2 
 
  CALL XGREND 
 
  STOP 
END PROGRAM HELICITY 
 
SUBROUTINE CIRCLE(x0, y0, r) 
  REAL :: x0, y0, r 
 
  deg2rad = atan(1.0)/45.0 
 
  x=x0+r*cos(0*deg2rad) 
  y=y0+r*sin(0*deg2rad) 
  call xpenup(x,y) 
 
  DO i=1,360 
    x=x0+r*cos(i*deg2rad) 
    y=y0+r*sin(i*deg2rad) 
    call xpendn(x,y) 
  ENDDO 
END SUBROUTINE CIRCLE 
Output of the program: 
 
 
u_storm =    8.485281     
v_strom =    8.485281     
 
     Z(km)   Dir(deg)   V(m/s)    ua(m/s)   va(m/s)    ur(m/s)  vr(i) 

     0.000   120.000     5.000    -4.330     2.500   -12.815    -5.985 
     1.000   150.000    10.000    -5.000     8.660   -13.485     0.175 
     2.000   180.000    15.000     0.000    15.000    -8.485     6.515 
     3.000   220.000    20.000    12.856    15.321     4.370     6.836 
     4.000   250.000    25.000    23.492     8.551    15.007     0.065 
     5.000   270.000    30.000    30.000     0.000    21.515    -8.485 
     6.000   310.000    40.000    30.642   -25.712    22.156   -34.197 
 
Layer omegax(1/s) omegay(1/s)  um(m/s)  vm(m/s) H(m/s**2)  RH    OmegaS 
1  -0.006160  -0.000670   -13.150    -2.905     0.083     0.994   0.006160 
2  -0.006340   0.005000   -10.985     3.345     0.086     0.932   0.007521 
3  -0.000321   0.012856    -2.057     6.675     0.086     0.963   0.012380 
4   0.006770   0.010637     9.689     3.450     0.102     0.789   0.009946 
5   0.008550   0.006508    18.261    -4.210     0.129     0.639   0.006870 
6   0.025712   0.000642    21.836   -21.341     0.548     0.697   0.017939 
 
   3km integrated helicity (m**2/s**2)=   255.798 
   BRN shear S (m/s) =    17.027 
   CAPE for BRN=10 is   1449.537 
   CAPE for BRN=45 is   6522.915 
 
Hodograph: 

(Note – the storm-relative velocity vectors should point from the 
‘convergence’ points – the tip of storm-motion vector, towards the notes) 

 
 
Question g: 
g). For this wind profile, what kind of CAPE values will give you a BRN that suggests a 

high probability of multicell and supercell storms, respectively? 
 
Multicell storms occur mostly when bulk Richardson number Rn = 2* CAPE/ (V6km-

VBL)2 > 45.   

2. (50%) The storm-relative environment helicity in the lowest 3 km layer is given as 
 

  
3

0
[( ) ]

km

Hkm
SREH V C dz  

 
  

 

which, based on definition ˆ 垐
H

dV dv du
k i j

dz dz dz
     




, can be rewritten as  

where ( )rV V C 
 

 is the storm-relative velocity. 
 
a).  Using the above information and your knowledge of analytic geometry, show that the 

SREH is equal to minus twice the signed (i.e., positive or negative) area swept out by 
the storm-relative wind vector between 0 and 3 km on a hodograph. Note that, by 
convention, an area is positive (negative) if it is swept out counterclockwise 
(clockwise).  

 Consider the SERH in a layer of depth dz in which  the wind vector increases from 

rV


to rV dV
 

, the SERH in the layer is 

 

 ˆ( ) rd SREH k V dV     
 

 

 
 The total SERH in the 3 km layer will be sum of the SERH in each of such layers. 
 
 In the above Figure, we can see that 
 

  ˆ( ) rd SREH k V dV     
 

  

  
ˆ ˆ( ) sin sin( )

sin( ) sin ' 2*

r r

r r r

k k V dV V dV

V dV V dV V h Area

 where 
2

rV h
Area  



. The negative sign is because of the definition of the area, 

which in this case is swept by rV


in the clockwise direction ( dV


 points to the right 

side of rV


). 



 7

 
 

 

3 3

0 0

ˆ ( )

2 _ _ _ _ _ _ 3 _

km km

rkm km

r

SREH k V dV d SREH

Total Area swepted by V in km depth

       

  

 
 

  

 
 
 To keep the problem simple, let's assume that wind observations are available at the 0 

and 3 km levels only (note – the above solution consider the general case where 

rV


varies continuously with height). 

 
b). If the storm-relative velocity at 0 and 3 km levels are 1 1 2 2( , ) ( , )r r r ru v and u v , 

respectively, show that SREH can be calculated from  
 

2 1 1 2r r r rSERH u v u v  . 
 

Hint: Think of how you calculated the storm-relative helicity in Problem 1, for each 

of those 6 layers. Also, rdV dV
 

because the storm motion vector is constant with 
height. 
 
The SERH in the 3 km layer is: 
 

[( ) ]HSREH V C dz  
 

 

c). Verify that for the following hodograph and a zero storm-motion vector, the above 

two methods give the same results. 
 
  
 
 
 
Just plug in the numbers. 
 
d)  Explain why larger SERH tends to promote longer lasting supercell storms? 
 

V(3km)=10m/s 

V(0km)=15m/s 

See notes. The key is that it leads to large correlation between w’ and ’, implying large 
vorticity in updraft – from our analysis on pressure perturbation associated with rotation 
updraft, we understand rotation in updraft produces additional positive lifting therefore 
stronger updraft therefore stronger storms. 


